
1 Introduction

To begin we will briefly review some results and terminology from the theory
of ordinary differential equations (ODEs). Consider an autonomous ODE

x′ = f(x), (1)

which admits an equilibrium point, x∗. The linearization of (1) about x∗ is
given by

x′ = Ax (2)

where A = Df(x∗). Recall that x∗ is called nonhyperbolic if at least one of
the eigenvalues of A has zero real part. Given a complete set of generalized
eigenvectors for the eigenvalues of A with zero real part, one can construct
a basis for the subspace solutions of (2) corresponding to these eigenvalues.
This subspace is called the centre eigenspace of (2). Nonhyperbolic equilibrium
points are important as they often occur at bifurcation points of a differential
equation. The centre manifold is a powerful tool for studying the behaviour of
solutions (and hence the nature of the bifurcation) of (1) in a neighbourhood of
a nonhyperbolic equilibrium point. It is a nonlinear manifold which is tangent
to the centre eigenspace at x∗. For a more detailed review of the theory and
construction of centre manifolds for ODES see (Guckenheimer and Holmes,
1983, Section 3.2), (Wiggins, 1990, Section 2.1) or (Perko, 1996, Section 2.12).

In this chapter we will study centre manifolds for nonhyperbolic equilib-
rium points of delay differential equations (DDEs). In general, one cannot find
the centre manifold exactly, thus one must construct an approximation. Some
authors have performed the construction by hand, e.g., Faria and Magalhães
(1995a,b); Gilsinn (2002); Guo and Huang (2003); Guo et al. (2004); Guo
(2005); Jiang et al. (2006); Kalmár-Nagy et al. (1999, 2001); Liu and Yuan
(2005); Nayfeh (2008); Orosz and Stépán (2004, 2006); Sri Namachchivaya

and van Roessel (2003); Wischert et al. (1994) and Wu et al. (1999). How-
ever, the construction generally involves a lot of computation and is most
easily accomplished either numerically or with the aid of a symbolic algebra
package. Here we focus on the symbolic algebra approach, which has been used
by several authors, e.g., Bélair and Campbell (1994); Campbell et al. (2005);
Qesmi et al. (2006b,a, 2007); Stone and Campbell (2004); Yuan et al. (2004);
Yuan and Campbell (2004); Yuan and Wei (2005) and Wei and Yuan (2005).
Unfortunately, there is rarely space in journal articles to give details of the
implementation of such computations. Thus, my purpose here is to give these
details, for a particular example DDE and for the symbolic algebra package
MapleTM, so that other authors may reproduce them in other contexts.

In the following section we will outline the theoretical setting for calculat-
ing centre manifolds. In the second section, we will show how the computations
may be implemented in the symbolic algebra package MapleTMby applying the
theory to a model due to Stone and Askari (2002). In the final section we will
discuss extensions of this approach as well as alternate approaches.

2 Theory

In this section we briefly outline the theoretical setting for calculating centre
manifolds. More detail on the theory can be found in Ait Babram et al. (1997),
Gilsinn (2008) and Hale and Verduyn Lunel (1993).

Consider the general delay differential equation

x′(t) = g(x(t),x(t− τ1), . . . ,x(t− τp);µ) (3)

where x ∈ IRn, g : IRn × IRn × . . . × IRn × IRk → IRn, p is a positive integer
and µ ∈ IRk and τj > 0, j = 1, . . . , p are parameters of the model. We shall
assume that g is as smooth as necessary for our subsequent computations (i.e.,
g ∈ Cr for r large enough) and the equation admits an equilibrium solution
x(t) = x∗. In general x∗ may depend on µ, but not on the τj . Shifting the
equilibrium to zero and separating the linear and nonlinear terms gives

x′(t) = A0(µ) x(t)+

p∑
j=1

Aj(µ) x(t−τj)+f(x(t),x(t−τ1), . . . ,x(t−τp);µ), (4)

where
Aj(µ) = Dj+1g(x∗, . . . ,x∗;µ) (5)

and

f(x(t),x(t− τ1), . . . ,x(t− τp);µ) = g(x(t),x(t− τ1), . . . ,x(t− τp);µ)
−A0(µ) x(t)−

∑p
j=1Aj(µ) x(t− τj).

(6)
Here Djg means the Jacobian of g with respect to its jth argument.

Let τ = maxj τj . To pose an initial value problem at t = t0 for this DDE
one must specify the value of x(t) not just at t = t0, but on the whole interval
[t0 − τ, t0]. Thus an appropriate initial condition is

x(t0 + θ) = ζ0(θ), −τ ≤ θ ≤ 0, (7)

where ζ0 : [−τ, 0)→ IRn is a given function. It can be shown (see e.g. (Hale and
Verduyn Lunel, 1993, Section 2.2) that if ζ0 is continuous and f is Lipschitz
there exists a unique solution to the initial value problem (4)–(7) which is
defined and continuous on a (maximal) interval [t0 − τ, β), β > 0. In the
following, we will assume that ζ0 and f satisfy these conditions.

To define an appropriate phase space for the solutions of the DDE, make
the following definition

xt(θ)
def
= x(t+ θ), −τ ≤ θ ≤ 0.

Note that the initial condition can now be expressed as xt0 = ζ0 and that xt
will be a continuous mapping from [−τ, 0]→ IRn for each t ∈ [t0, β).

With this in mind, it is usual (see, e.g., Hale and Verduyn Lunel (1993))

to take the phase space for (4) to be the Banach space C def
= C([−τ, 0], IRn)

of continuous mappings from [−τ, 0] into IRn, equipped with the norm

‖ζ‖τ = sup
θ∈[−τ,0]

‖ζ(θ)‖,

where ‖ · ‖ is the usual Euclidean norm on IRn. We can then define the flow
for the DDE as a mapping on C which takes the initial function ζ0 into the
function xt.

The equation (4) for x(t) can be expressed as a functional differential
equation (FDE)

x′(t) = L(xt;µ) + F(xt;µ), (8)

where L : C × IRk → IRn is a linear mapping defined by

L(φ;µ) = A0(µ)φ(0) +

p∑
j=1

Aj(µ)φ(−τj), (9)

and F : C × IRk → IRn is a nonlinear functional defined by

F(φ;µ) = f(φ(0), φ(−τ1), . . . , φ(−τp);µ). (10)

As shown in e.g., Faria and Magalhães (1995a,b) and Wischert et al. (1994)
one may extend (8) to a differential equation for xt(θ) as follows

d

dt
xt(θ) =


d
dθ (xt(θ)) , −τ ≤ θ < 0

L(xt;µ) + F(xt;µ) , θ = 0 ,
. (11)

This equation will be important for the centre manifold construction.

2.1 Linearization

Clearly (4) and (8) admit the trivial solution x(t) = 0, ∀t, which corresponds
to the equilibrium solution x(t) = x∗ of (3). The stability of this equilibrium
solution can be studied via the linearization of (8) about the trivial solution:

x′(t) = L(xt;µ) (12)

or, in the DDE form,

x′(t) = A0(µ) x(t) +

p∑
j=1

Aj(µ) x(t− τj). (13)

Substituting the ansatz x(t) = eλtv, v ∈ IRn into (13) yields the matrix
vector equation λI −A0(µ)−

p∑
j=1

Aj(µ)e−λτj

v = 0, (14)

which we will sometimes write in the compact form ∆(λ;µ)v = 0. Requiring
nontrivial solutions (v 6= 0) yields the constraint det(∆(λ);µ) = 0, i.e., that
λ is a root of the characteristic equation

det

λI −A0(µ)−
p∑
j=1

Aj(µ)e−λτj

 = 0. (15)

It can be shown (Hale and Verduyn Lunel, 1993, Corollary 7.6.1) that the
trivial solution of (12) (or (13)) will be asymptotically stable (and hence the
equilibrium solution of (33) will be locally asymptotically stable) if all the
roots of (15) negative real parts. We will call these roots the eigenvalues of
the equilibrium point.

Consider a point, µ = µc, in the parameter space where the characteristic
equation (15) has m roots with zero real parts and the rest of the roots have
negative real parts. The following results are shown in Hale and Verduyn
Lunel (1993). At such a point there exists a decomposition of the solution
space for the linear FDE (12) as C = N ⊕ S, where N is an m-dimensional
subspace spanned by the solutions to (12) corresponding to the eigenvalues
with zero real part, S is infinite dimensional and N and S are invariant under
the flow associated with (12). N and S are analogous to the centre and stable
eigenspaces for ODEs.

For simplicity, we will assume that all the eigenvalues with zero real part
have multiplicity one. This includes the most common cases studied: single
Hopf bifurcation, double Hopf bifurcation (with nonidentical frequencies) and
zero-Hopf bifurcation. For a discussion of DDEs with a zero eigenvalue of

multiplicity two (Bogdanov Takens bifurcation) or three, see Campbell and
Yuan (2008) or Qesmi et al. (2007). For a discussion of DDEs where com-
plex conjugate eigenvalues with higher multiplicity arise due to symmetry,
see Campbell et al. (2005); Guo and Huang (2003); Guo et al. (2004); Guo
(2005); Krawcewicz and Wu (1999); Orosz and Stépán (2004); Wu (1998);
Wu et al. (1999) or Yuan and Campbell (2004). For a general discussion of
eigenspaces associated with eigenvalues of higher multiplicity in DDEs see
(Hale and Verduyn Lunel, 1993, Section 7.4).

Let {φ1(t), φ2(t), . . . , φm(t)} be a basis for N and {λ1, λ2, . . . , λm} the
corresponding eigenvalues. This basis can be constructed in a similar manner
to that for ODEs. Since Re(λk) = 0 for each k, either λk = 0 or λk = iωk. In
the latter case, it is easy to check that −iωk is also a root, and we will order the
eigenvalues so that λk+1 = −iωk. With the restriction of simple eigenvalues,
the construction of the basis functions is straight forward. If λk = 0, then φk =
vk where vk a solution of ∆(0;µc)vk = 0. If λk = iω, then φk = Re(eiωk tvk)
and φk+1 = Im(eiωk tvk), where vk a solution of ∆(iωk;µc)vk = 0. In the
following, we will usually write the basis as an n × m matrix, with the kth

column given by φk, viz.:

Φ(t) = [φ1(t) |φ2(t) | . . . |φm(t) ] . (16)

A simple calculation then shows that Φ satisfies the following matrix ordinary
differential equation:

Φ′ = ΦB (17)

where B is a block diagonal matrix, with block [ 0 ] for the zero eigenvalue, if
present, and block [

0 ωk
−ωk 0

]
for each pair of complex conjugate eigenvalues, ±iωk.

Note that the basis functions may also be treated as functions on C, by
changing their argument to θ ∈ [−τ, 0]. Now consider

L(eλkθvk;µc) = A0(µc)vk +

p∑
j=1

Aj(µc)e
−λkτjvk

= λkvk,

which follows from (14). This implies that L(φk;µc) = 0 when λk = 0, and
when λk = iωk, L(φk;µc) = −ωkφk+1(0) and L(φk+1;µc) = ωkφk(0) . We
then have the following result:

L(Φ;µc) = Φ(0)B. (18)

As for ODEs, the decomposition of the solution space may be accomplished
via the introduction of the adjoint equation for (12). However, a different
equation, which is closely related to the adjoint equation, may also be used to

decompose the solution space. It turns out that this latter equation is useful
for the centre manifold construction, so we focus on it.

Let Rn∗ be the n-dimensional row vectors and C∗ = C([0, r], IRn∗). For
ψ ∈ C∗ and φ ∈ C, define the following bilinear form

〈ψ, φ〉 =

n∑
i=1

ψj(0)φj(0) +

p∑
j=1

∫ 0

−τj
ψ(σ + τj)Aj φ(σ)dσ. (19)

As shown in (Hale and Verduyn Lunel, 1993, Section 7.5), this can be used
to define a system dual to (12) given by

y′(t) = LT (ys;µ), s ≤ 0, (20)

where ys = y(s+ ξ), 0 ≤ ξ ≤ τ and LT is a linear mapping on C∗× IRk given
by

LT (ψ;µ) = −ψ(0)A0(µ)−
p∑
j=1

ψ(τj)Aj(µ). (21)

Equation (20) is called the transposed system by Hale and Verduyn Lunel
(1993). In the literature, it is sometimes called the formal adjoint. The corre-
sponding differential equation is

y′(s) = −y(s)A0(µ)−
p∑
j=1

y(s+ τj)Aj(µ), s ≤ 0. (22)

Using the ansatz y(s) = we−λ s, w ∈ IRn∗ and proceeding as for (12),
shows that w must satisfy w∆(λ;µ) = 0. Thus the characteristic equation of
(22) is just (15). It follows that the trivial solutions of (22) and (13) have the
same eigenvalues.

Let

Ψ(s) =

 ψ1(s)
...

ψm(s)

 ,
be a basis for the solutions of (20) (or, equivalently, (22)) corresponding to
the m eigenvalues with zero real part (i.e. the “centre eigenspace” of (22)).
Note that the ψj are row vectors and that they can be considered as functions
on C∗ if we change their argument to ξ ∈ [0, τ ]. The fundamental result used
in the centre manifold construction is that Ψ may be used to decompose the
solution space. See (Hale and Verduyn Lunel, 1993, Section 7.5) for details
and proofs. In particular, for any ζ ∈ S,

〈ψj , ζ〉 = 0, j = 1, . . . ,m.

Further, we can choose a basis so that 〈Ψ,Φ〉 = I, where 〈Ψ,Φ〉 is the m×m
matrix with i, j elements 〈ψi, φj〉 and I is the m ×m identity matrix. Thus

for any ζ ∈ N we have ζ = Φ u where u = 〈Ψ, ζ〉 ∈ IRm. Finally, one can
show that

Ψ′ = −BΨ and LT (Ψ;µc) = −BΨ(0). (23)

where B is the same block diagonal matrix as in (17).

2.2 Nonlinear Equation

Now let us return to the nonlinear equation (8). For the rest of this section
we will assume that µ = µc and hence that the characteristic equation (15)
has m eigenvalues with zero real parts and all other eigenvalues have negative
real parts. In this situation Hale and Verduyn Lunel (1993) (Chapter 10) have
shown that there exists, in the solution space C for the nonlinear FDE (8), an
m dimensional centre manifold. Since all the other eigenvalues have negative
real parts, this manifold is attracting and the long term behaviour of solutions
to the nonlinear equation is well approximated by the flow on this manifold.
In particular, studying the flow on this manifold will enable us to characterize
the bifurcation which occurs as a µ passes µc. Below, we outline the steps
involved in computing this manifold. The approach we take follows the work
of Hale (1985) and Wischert et al. (1994) (scalar case) and of Ait Babram
et al. (1997) (vector case). Since all our computations will be done for µ = µc,
we not write the dependence on µ explicitly.

To begin, we note that points on the local centre manifold of 0 can be
expressed as the sum of a linear part belonging to N and a nonlinear part
belonging to S, i.e.,

W c
loc(0) = {φ ∈ C | φ = Φu + h(u)} ,

where Φ(θ), θ ∈ [−τ, 0] is the basis forN introduced above, u ∈ IRm, h(u) ∈ S
and ‖u‖ is sufficiently small. The solutions of (8) on this centre manifold are
then given by x(t) = xt(0) where xt(θ) is a solution of (11) satisfying

xt(θ) = Φ(θ)u(t) + h(θ,u(t)) . (24)

To find the center manifold and the solutions on it, we proceed as follows.
Substituting (24) into (11) yields

[
Φ(θ) +

∂h

∂u

]
u̇(t) =


Φ′(θ)u(t) +

∂h

∂θ
, −τ ≤ θ < 0

L(Φ(θ))u(t) + L(h(θ,u(t)))
+F[Φ(θ)u(t) + h(θ,u(t))] , θ = 0 .

(25)

Using (17) and (18) in (25) we obtain

[
Φ(θ) +

∂h

∂u

]
u̇(t) =


Φ(θ)Bu(t) +

∂h

∂θ
, −τ ≤ θ < 0,

Φ(0)Bu(t) + L(h(θ,u(t)))
+F[Φ(θ)u(t) + h(θ,u(t))] , θ = 0.

(26)

This coupled system must be solved for u(t) and h(θ,u(t)).
To derive the equation for u(t) we will use the bilinear form (19). First we

note some useful results. Since h(θ,u) ∈ S for any u,

〈Ψ(ξ),h(θ,u(t))〉 = 0.

It then follows from the definition of the partial derivative that

〈Ψ(ξ),
∂h

∂u
(θ,u(t))〉 = 0.

Finally, using (23) we have

Ψ(0)L(h(θ,u)) +

p∑
j=1

∫ 0

−τj
Ψ(σ + τj)Aj

∂h

∂σ
dσ

= −LT (Ψ(ξ))h(0,u(t))−
p∑
j=1

∫ 0

−τj
Ψ′(σ + τj)Aj h(σ,u(t)) dσ

= BΨ(0)h(0,u(t)) +

p∑
j=1

∫ 0

−τj
BΨ(σ + τj)Aj h(σ,u(t)) dσ

= B〈Ψ(ξ),h(θ,u)〉
= 0.

Applying the bilinear form to Ψ and (26) and using these results gives the
following system of ODEs for u(t):

u̇(t) = Bu(t) + Ψ(0)F[Φ(θ)u(t) + h(θ,u(t))] . (27)

Using (27) in (26) then yields the following system of partial differential equa-
tions for h(θ,u):

∂h

∂u
{Bu + Ψ(0)F[Φ(θ)u + h(θ,u)]}+ Φ(θ)Ψ(0)F[Φ(θ)u + h(θ,u)]

=

{
∂h

∂θ
, −τ ≤ θ < 0

L(h(θ,u)) + F[Φ(θ)u + h(θ,u)] , θ = 0 .
(28)

Thus, the evolution of solutions on the centre manifold is determined by solv-
ing (28) for h(θ,u) and then (27) for u(t). To solve (28), one uses a standard
approach in centre manifold theory, namely, one assumes that h(θ,u) may be
expanded in power series in u:

h(θ,u) = h2(θ,u) + h3(θ,u) + · · · (29)

where

h2(θ,u) =

h
1
11(θ)u21 + · · ·h11m(θ)u1um + h122(θ)u22 + · · ·+ h1mm(θ)u2m

...
hn11(θ)u21 + · · ·hn1m(θ)u1um + hn22(θ)u22 + · · ·+ hnmm(θ)u2m

 ,
and similarly for h3 and the higher order terms.

Before proceeding to solve for h, we would like to note that to deter-
mine the terms of (27) to O(‖u(t)‖l), one only needs the terms which are
O(‖u(t)‖l−1) in the series for h. To see this, write F in series form

F = F2 + F3 + · · · (30)

and hence rewrite (27) as

u̇ = Bu + Ψ(0)
{
F2[Φ(θ)u + h2(θ,u) + h3(θ,u) +O(‖u‖4)]

+F3[Φ(θ)u + h2(θ,u) + h3(θ,u) + O(‖u‖4)] +O(‖u‖4)
}
.

Expanding each Fj in a Taylor series about Φ(θ)u yields

u̇ = Bu + Ψ(0) [F2(Φ(θ)u) +DF2(Φ(θ)u)h2(θ,u) + F3(Φ(θ)u)] +O(‖u‖4).

Thus we see that h2 is only needed to calculate the third order terms not
the second order terms. A similar result holds for the higher order terms.
Of particular note is the fact that if the lowest order terms we need in the
centre manifold are the same as the lowest order terms in F, then there is
no need to calculate h at all! This is the case for a Hopf bifurcation when
F2 = 0. Examples of this can be found in Wischert et al. (1994); Stépán and
Haller (1995) and Landry et al. (2005). This is also the case when the normal
form for a particular bifurcation is determined at second order, such as for
the Bogdanov-Takens bifurcation (see, e.g., Campbell and Yuan (2008)) or a
double Hopf bifurcation with 1:2 resonance (see, e.g., Campbell and LeBlanc
(1998)).

Now let us return to solving (28). Substituting (29) and (30) into the first
part of (28) and expanding the Fj about Φ(θ)u yields

∂h2

∂θ
+O(‖u‖3) =

∂h2

∂u
(θ,u)Bu + Φ(θ)Ψ(0)F2(Φ(θ)u) +O(‖u‖3). (31)

Equating terms with like powers of u1, . . . , um in this equation yields a system
of ODEs for the hijk(θ). The system is linear and is easily solved to find the

general solutions for the hijk(θ) in terms of arbitrary constants.
These arbitrary constants may be determined as follows. Substituting (29)

and (30) into the second part of (28) and expanding the Fj about Φ(θ)u yields

∂h2

∂u

∣∣∣∣
θ=0

Bu + Φ(0)Ψ(0)F2(Φ(θ)u) +O(‖u‖3)

= L(h2(θ,u)) + F2(Φ(θ)u) +O(‖u‖3)
(32)

Equating terms with like powers of u1, . . . , um in this equation yields a set of
boundary conditions for the arbitrary constants.

Once one has determined h2 one may proceed to the next order of approx-
imation and calculate h3. As discussed above, however, for most applications
this is unnecessary.

3 Application

Now consider the model of Stone and Askari (2002):

η′′ + δη′ + η − β(1− µ(η − η(t− τ)))(p0 + p1η
′ + p2η

′2) = 0 (33)

which was developed to study the vibrations in drilling. This model is in
dimensionless form, the model in physical variables can be found in Stone
and Askari (2002) or Stone and Campbell (2004). The variable η corresponds
to the amplitude of the vibrations and ′ to derivative with respect to time.
The parameter 1/τ is proportional to the speed of rotation of the drill and
β to the width of cut. Since these two parameters can be varied in practice,
Stone and Campbell (2004) chose these as the bifurcation parameters. The
other parameters, µ, p0, p1, p2, can be related to other physical parameters
(see Stone and Campbell (2004)).

Note that this equation has an equilibrium solution η(t) = βp0. which
corresponds to the steady cutting solution. The drilling process may exhibit
chatter which is a self excited oscillation of the drill. The emergence of chatter
in the physical system corresponds to a Hopf bifurcation in the model (33). In
Stone and Campbell (2004) the criticality of this bifurcation was studied using
the centre manifold construction described in section 1. We will reproduce the
essence of the analysis here, including the relevant commands in MapleTM113

used to perform the computations symbolically. Commands will be written in
typewriter font and preceded by a >. Each command will be followed by the
output produced when it is executed. If a command ends with a colon then
no output is printed. More information on Maple can be found in the manual:
Maple User Manual (2006).

Shifting the equilibrium to the origin and rewriting the equation as a first
order vector equation puts it in the form (4):

x′(t) = A0x(t) +A1x(t− τ) + f(x(t),x(t− τ)) (34)

where

x(t) =

[
x(t)
x′(t)

]
=

[
η(t)− β p0

η′(t)

]
(35)

A0 =

[
0 1

−1− β p0 µ β p1 − δ

]
, A1 =

[
0 0

β p0 µ 0

]
, (36)

3 The commands used are backward compatible to at least MapleTM9.5.

and

f =

[
0

β p2x
′(t)2 − β µ(x(t)− x(t− τ))(p1x

′(t) + p2x
′(t)2)

]
. (37)

Thus, in this example, the linear mapping of eq. (8) is given by

L(φ(θ)) = A0φ(0) +A1φ(−τ), (38)

and F : C → IR2 is a nonlinear functional defined by

F(φ(θ)) = f(φ(0), φ(−τ)). (39)

The characteristic matrix and equation for this example can be defined in
Maple as follows.

> A0:=matrix(2,2,[[0,1],[-1-beta*p0*mu,beta*p1-delta]]);

A0 :=

[
0 1

−1− β p0µ β p1− δ

]
> A1:=matrix(2,2,[[0,0],[beta*p0*mu,0]]);

A1 :=

[
0 0

β p0µ 0

]
> ident:=evalm(array(1..2,1..2,identity));

ident :=

[
1 0
0 1

]
> Delta:=evalm(lambda*ident-A0-exp(-lambda*tau)*A1);

∆ :=

[
λ −1

1 + β p0µ− β p0µe−λ τ λ− β p1 + δ

]
> char_eq:=collect(det(Delta),lambda);

char eq := λ2 + (−β p1 + δ)λ+ 1 + β p0µ− e(−λ τ)β p0µ

The work of Stone and Campbell (2004) described curves, in the τ, β pa-
rameter space, along which the equilibrium solution of eq. (33) loses stability.
At each point on these curves, the characteristic equation (15) has a pair of
pure imaginary roots and the rest of the roots have negative real parts. Equa-
tions describing where the characteristic equation has pure imaginary roots
can be easily found in Maple:

> eq_im:=evalc(subs(lambda=I*omega,char_eq)):

eq_Re:=coeff(eq_im,I,0);

eq_Im:=coeff(eq_im,I,1);

eq Re := −ω2 + 1 + β p0µ− cos(ω τ)β p0µ

eq Im := −ω β p1 + ω δ + sin(ω τ)β p0µ

Stone and Campbell (2004) solved these equations to find expressions for τ
and β in terms of ω and the other parameters. For fixed values of the other
parameters, these expressions determined curves in the τ, β parameter space
which are parametrized by ω.

It is straightforward to check that the DDE (33) satisfies the conditions
for a Hopf bifurcation to occur as one passes through a point on these curves
(see (Hale and Verduyn Lunel, 1993, pp. 331-333) or (Gilsinn, 2002, Section
2) for a statement of the Hopf bifurcation Theorem for FDE’s). To determine
the criticality of this Hopf bifurcation, we compute the centre manifold of
the equilibrium point at the Hopf bifurcation, following the steps outlined in
section 2.

To begin, we calculate a basis for the “centre eigenspace”, N . To do this
we need to find the eigenfunctions corresponding to the eigenvalues ±iω. It
Maple this may be done as follows. First solve ∆(iω)v = 0 where ∆(iω) is the
characteristic matrix with λ = iω.

> v:=matrix([[v1],[v2]]);

Dv:=subs(lambda=I*omega,evalm(multiply(Delta,v)));

v2res:=v2=solve(Dv[1,1], v2);

v :=

[
v1
v2

]

Dv :=

[
I ω v1− v2(

1 + β p0µ− e−I ω τβ p0µ
)
v1 + (I ω − β p1 + δ) v2

]
v2res := v2 = I ω v1

Then define the complex eigenfunction and take the real and imaginary parts.

> yy:=map(evalc,subs(v2res,v1=1,evalm(exp(I*omega*theta)*v))):

Phi:=array(1..2,1..2,[[coeff(yy[1,1],I,0),coeff(yy[1,1],I,1)],

[coeff(yy[2,1],I,0),coeff(yy[2,1],I,1)]]);

Φ :=

[
cos(ωθ) sin(ωθ)
− sin(ωθ)ω cos(ωθ)ω

]
Similarly we define u and Φ u.

u:=matrix([[u1],[u2]]);

u :=

[
u1
u2

]
Phiu:=multiply(Phi,u);

Phiu :=

[
cos (ω θ) u1 + sin (ω θ) u2
− sin(ω θ)ω u1 + cos(ω θ)ω u2

]
Next define the matrix B.

> B:=matrix([[0,omega],[-omega,0]]);

B :=

[
0 ω
−ω 0

]
.

We need to define the basis, Ψ(ξ), ξ ∈ [0, τ ], for the “centre eigenspace”
of the transpose system. First we calculate a general basis Ψg in the same
way as we set up the basis Φ.

> w:=array(1..2):

wD:=subs(lambda=I*omega,multiply(w,Delta)):

w1res:=w[1]=solve(wD[2],w[1]):

yy:=map(evalc,subs(w1res,w[2]=1,lambda=I*omega,

evalm(w*exp(-lambda*xi)))):

Psi_g:=array(1..2,1..2,[[coeff(yy[1],I,0),coeff(yy[2],I,0)],

[coeff(yy[1],I,1),coeff(yy[2],I,1)]]);

Ψg :=

[
cos(ω ξ)(−βp1 + δ) + sin(ω ξ)ω cos(ω ξ)
− sin(ω ξ)(−βp1 + δ) + cos(ω ξ)ω − sin(ω ξ)

]
We now wish to find a basis Ψ such that 〈Ψ,Φ〉 = I. The elements of Ψ will
be linear combinations of those of Ψg, i.e., Ψ = KΨg where K is a 2 × 2
matrix of constants. Thus we have

I = 〈Ψ,Φ〉
= 〈KΨg,Φ〉
= K〈Ψg,Φ〉

Which implies that K = 〈Ψg,Φ〉−1.
For this example the bilinear form (19) becomes

〈ψ, φ〉 =

2∑
j=1

ψj(0)φj(0) +

∫ 0

−τ
ψ(σ + τ)A1φ(σ) dσ

= ψ(0) · φ(0) + β µ p0

∫ 0

−τ
ψ2(σ + τ)φ1(σ) dσ

We define this bilinear form as a procedure as follows:

> bilinear_form:=proc(rowv,colv)

local pstemp;

pstemp:=subs(xi=0,theta=0,innerprod(rowv,colv))

+int(subs(xi=sigma+tau,theta=sigma,

innerprod(rowv,A1,colv)),sigma=-tau .. 0);

RETURN(pstemp)

end:

Note that the command innerprod calculates the dot product when given
two vectors and the vector-matrix-vector product when given two vectors and
a matrix.

Next we apply the bilinear form to each row of Ψg and each column of
Φ and store the result in the matrix produit. We then invert produit and
multiply the result by Ψg. 4

> rowvec:=array(1..2): colvec:=array(1..2): produit:=matrix(2,2):

> for I1 from 1 to 2 do

for I2 from 1 to 2 do

rowvec:=row(Psi_g,I1);

colvec:=col(Phi,I2);

produit[I1,I2]:=eval(bilinear_form(rowvec,colvec));

od;

od;

> K:=inverse(produit):

> PPsi:=map(simplify,multiply(K,Psi_g)):

In fact, all we need for subsequent calculations is Ψ(0). To keep the expressions
from getting too large we will define an empty matrix Ψ0 to use as a place
holder. We will store the actual values of Ψ(0) in the list Psi0 vals.

Psi0:=matrix(2,2);

Psi0_res:=map(simplify,map(eval,subs(xi=0,evalm(PPsi)))):

Psi0_vals:=[Psi0[1,1]=Psi0_res[1,1],Psi0[1,2]=Psi0_res[1,2],

Psi0[2,1]=Psi0_res[2,1],Psi0[2,2]=Psi0_res[2,2]]:

Ψ0 = array(1 .. 2, 1 .. 2, [])

Now, to determine the criticality of the Hopf bifurcation, one need only
find the terms up to and including those which are O(‖u(t)‖3) in (27). Thus,
as discussed in the previous section, we only need the quadratic terms in the
series for h. We thus define

h:=matrix([[h1_11(theta)*u1^2+h1_12(theta)*u1*u2

+h1_22(theta)*u2^2],

[h2_11(theta)*u1^2+h2_12(theta)*u1*u2+h2_22(theta)*u2^2]]);

4 Note that Psi is a reserved word, so we use PPsi instead.

h :=

[
h1 11(θ)u12 + h1 12(θ)u1u2 + h1 22(θ)u22

h2 11(θ)u12 + h2 12(θ)u1u2 + h2 22(θ)u22

]
(40)

We define the linear and nonlinear parts of the DE as follows

> x:=matrix([[x1],[x2]]);

xt:=matrix([[x1t],[x2t]]);

lin:=evalm(multiply(A0,x)+multiply(A1,xt));

f:= beta*p2*x2^2-beta*mu*p1*x1*x2+beta*mu*p1*x1t*x2

-beta*mu*p2*x1*x2^2+beta*mu*p2*x1t*x2^2:

nonlin:=matrix([[0],[f]]);

x :=

[
x1
x2

]

xt :=

[
x1t
x2t

]
lin :=

[
x2

(−1− β µ p0)x1 + (β p1− δ)x2 + β µ p0x1t

]
nonlin :=

[
0

β p2x22 − βµ p1x1x2 + β µ p1x1t x2− β µ p2x1x22 + β µ p2x1t x22

]
Then we define the expressions, in terms of the coordinates u, for points on
the centre eigenspace, x ce, on the centre manifold, x cm, and on the nonlinear
terms of the centre manifold, x h.

> Phiu0:=map(eval,subs(theta=0,evalm(Phiu))):

Phiut:=map(eval,subs(theta=-tau,evalm(Phiu))):

x_ce:=[x1=Phiu0[1,1],x2=Phiu0[2,1],x1t=Phiut[1,1],

x2t=Phiut[2,1]];

h0:=map(eval,subs(theta=0,evalm(h))):

ht:=map(eval,subs(theta=-tau,evalm(h))):

x_cm:=[x1=Phiu0[1,1]+h0[1,1],x2=Phiu0[2,1]+h0[2,1],x1t=Phiut[1,1]+ht[1,1],

x2t=Phiut[2,1]+ht[2,1]];

x_h:=[x1=h0[1,1],x2=h0[2,1],x1t=ht[1,1],x2t=ht[2,1]]:

x ce := [x1 = u1, x2 = ω u2, x1t = cos(ω τ)u1− sin(ω τ)u2,
x2t = sin(ω τ)ω u1 + cos(ω τ)ω u2]

x cm := [x1 = u1 + h1 11(0)u12 + h1 12(0)u1u2 + h1 22(0)u22,
x2 = ω u2 + h2 11(0)u12 + h2 12(0)u1u2 + h2 22(0)u22,
x1t = cos(ω τ)u1− sin(ω τ)u2 + h1 11(−τ)u12

+h1 12(−τ)u1u2 + h1 22(−τ)u22,
x2t = sin(ω τ)ω u1 + cos(ω τ)ω u2 + h2 11(−τ)u12

+h2 12(−τ)u1u2 + h2 22(−τ)u22]

We can now define differential equations for the hijk. First define the left
hand side of (31).



16 Sue Ann Campbell

> delhs:=map(diff,h,theta);

delhs :=


(
d

dθ
h1 11(θ)

)
u12 +

(
d

dθ
h1 12(θ)

)
u1u2 +

(
d

dθ
h1 22(θ)

)
u22(

d

dθ
h2 11(θ)

)
u12 +

(
d

dθ
h2 12(θ)

)
u1u2 +

(
d

dθ
h2 22(θ)

)
u22


Now define the right hand side of (31).

> dhdu:=matrix([[diff(h[1,1],u1),diff(h[1,1],u2)],

[diff(h[2,1],u1),diff(h[2,1],u2)]]);

dhdu :=

[
2h1 11(θ)u1 + h1 12(θ)u2 h1 12(θ)u1 + 2h1 22(θ)u2
2h2 11(θ)u1 + h2 12(θ)u2 h2 12(θ)u1 + 2h2 22(θ)u2

]
> derhs:=map(collect,map(expand,evalm(multiply(dhdu,multiply(B,u))+

multiply(Phi,multiply(Psi0,[0,subs(x_ce,f)])))),

[u2,u2],distributed,factor):

The expression for derhs is quite long, so we don’t display it. Now we put
together the right hand side and left hand side. The coefficient of each distinct
monomial, u1ku2j , j+k = 2, determines one differential equation. We display
two of them as examples.

> hdes:=delhs-derhs:

de1:=coeff(coeff(hdes[1,1],u1^2),u2,0);

de2:=coeff(coeff(hdes[1,1],u1),u2);

de3:=coeff(coeff(hdes[1,1],u2^2),u1,0):

de4:=coeff(coeff(hdes[2,1],u1^2),u2,0):

de5:=coeff(coeff(hdes[2,1],u1),u2):

de6:=coeff(coeff(hdes[2,1],u2^2),u1,0):

de1 :=

(
d

dθ
h1 11(θ)

)
+ ω h1 12(θ)

de2 :=

(
d

dθ
h1 12(θ)

)
− 2ω (h1 11(θ)− h1 22(θ)) + ω β µ p1(1− cos(ωτ))

(cos(ωθ) Ψ1,2(0) + sin(ωθ) Ψ2,2(0))

Now define the list of differential equations and functions to solve for.

> des:={de1,de2,de3,de4,de5,de6}:

fns:={coeff(h[1,1],u1^2),coeff(coeff(h[1,1],u1),u2),

coeff(h[1,1],u2^2), coeff(h[2,1],u1^2),

coeff(coeff(h[2,1],u1),u2),coeff(h[2,1],u2^2)};

fns := {h1 11(θ), h1 12(θ), h1 22(θ), h2 11(θ), h2 12(θ), h2 22(θ)}

The DEs are linear and are easily solved to find the general solutions for the
hijk(θ) in terms of six arbitrary constants using the command dsolve. For
convenience, we rename the arbitrary constants.

> temp:=dsolve(des,fns):

changeC:=[_C1=C1,_C2=C2,_C3=C3,_C4=C4,_C5=C5,_C6=C6];

hsoln:=simplify(expand(evalc(subs(changeC,value(temp))))):

The solutions are quite long, so we show only one example.

> collect(hsoln[6],[Psi0[1,1],Psi0[1,2],Psi0[2,1],Psi0[2,2],

p1,p2],factor);

h1 22(θ) =

(
−1

3
β µ (cos(ωθ) + sin(ωθ) sin(ωτ)− cos(ωθ) cos(ωτ)) p1

+
1

3
β sin(ωθ)ω p2

)
Ψ01,2 +

(
1

3
β µ(cos(ωθ) sin(ωτ)

− sin(ωθ) + sin(ωθ) cos(ωτ)) p1 −1

3
β cos(ωθ)ω p2

)
Ψ02,2

−C6 cos(ωθ)2 + C2 +
1

2
C6 + C5 sin(ωθ) cos(ωθ)

Recall that the values for Ψ0i,j are stored in the list Psi0 vals. Later we will
need the values of hijk(0) and hijk(−τ), so we store them in the sets hsoln0

and hsolnt.

> hsoln0:=simplify(eval(subs(theta=0,hsoln))):

hsolnt:=simplify(eval(subs(theta=-tau,hsoln))):

We now set up the boundary conditions to solve for the arbitrary constants,
C1, C2, . . .. Note that the left hand side of (32) is just the right hand side of
(31) with θ = 0.

> bclhs:=map(eval,subs(theta=0,evalm(derhs))):

bcrhs:=map(collect,evalm(subs(x_h,evalm(lin))+

subs(x_ce,evalm(nonlin))),[u1,u2]);

Now we put together the right hand side and left hand side. The coefficient
of each distinct monomial, u1ku2j , j + k = 2, determines one boundary
condition.

> consts:=[C1,C2,C3,C4,C5,C6];

bceq:=subs(hsoln0,hsolnt,evalm(bclhs-bcrhs)):

bc1:=collect(coeff(coeff(bceq[1,1],u1,2),u2,0),consts);

bc2:=collect(coeff(coeff(bceq[1,1],u1,1),u2,1),consts):

bc3:=collect(coeff(coeff(bceq[1,1],u1,0),u2,2),consts):

bc4:=collect(coeff(coeff(bceq[2,1],u1,2),u2,0),consts);

bc5:=collect(coeff(coeff(bceq[2,1],u1,1),u2,1),consts):

bc6:=collect(coeff(coeff(bceq[2,1],u1,0),u2,2),consts):

Form the list of boundary conditions and solve using solve.

> bcs:={bc1,bc2,bc3,bc4,bc5,bc6}:

consts:=convert(const,set);

Csoln:=map(simplify,solve(bcs,consts)):

The solutions are quite long, so we show only one example.

> collect(Csoln[2],[Psi0[1,1],Psi0[1,2],Psi0[2,1],Psi0[2,2],

p1,p2],factor);

C2 :=

(
−1

2
β2ω µ sin(ω τ)p12 +

(
1

2
β2ω2 p2 +

1

2
µ (δ ω sin(ω τ) + β p0µ

−β p0µ cos(ω τ)2 )β
)
p1− 1

2
ω (β p0µ sin (ω τ) + δ ω)β p2

)
Ψ01,2

+

(
1

2
µ sin(ω τ)

(
−1− β p0µ+ β p0µ cos(ω τ) + ω2

)
β p1

−1

2
ω
(
−1− β p0µ+ β p0µ cos(ω τ) + ω2

)
β p2

)
Ψ02,2

−1

2
β p1µ sin(ω τ)ω +

1

2
β ω2 p2

The final step is to use the expressions for Ψ(0), Φ and h to calculate the
nonlinear terms of (27).

> fu:=collect(subs(x_cm,f),[u1,u2],distributed,factor);

nonlinu:=matrix([[0],[fu]]):

ODE_nonlin:=multiply(Psi0,nonlinu):

Note that we have used the fact that the first component of the nonlinearity
in our example equation (34) is 0.

Recalling our expression for the matrix B, we can see that for our example,
the general equation on the centre manifold (27) becomes (to O(‖u‖3))

u̇1 = ωu2 + f111u
2
1 + f112u1u2 + f122u

2
2 + f1111u

3
1 + f1112u

2
1u2

+f1122u1u
2
2 + f1222u

3
2,

u̇2 = −ωu1 + f211u
2
1 + f212u1u2 + f222u

2
2 + f2111u

3
1 + f2112u

2
1u2

+f2122u1u
2
2 + f2222u

3
2.

(41)

The f ijk and f ijkl are functions of the parameters β, τ, δ, θ, p0, p1, p2, the Hopf

frequency ω, and the centre manifold coefficients hijk(0) and hijk(−τ). As
should be expected, eq. (41) is an ODE at a Hopf bifurcation. The criticality

of this bifurcation (and hence of the Hopf bifurcation in the original system of
DDE’s) may be determined by applying standard approaches. For example,
one can show that the criticality of the Hopf bifurcation of (41) is determined
by the sign of the quantity (Guckenheimer and Holmes, 1983, p. 152)

a = 1
8

(
3f1111 + f1122 + f2112 + 3f2222

)
− 1

8ω

(
f112(f111 + f122)− f212(f211 + f222)− 2f111f

2
11 + 2f122f

2
22

)
.

(42)

To evaluate this expression, we first find the coefficients of the quadratic
terms.

> quad:=array(1..2,1..3):

> quad[1,1]:=coeff(coeff(ODE_nonlin[1,1],u1,2),u2,0);

> quad[1,2]:=coeff(coeff(ODE_nonlin[1,1],u1,1),u2,1);

> quad[1,3]:=coeff(coeff(ODE_nonlin[1,1],u1,0),u2,2);

> quad[2,1]:=coeff(coeff(ODE_nonlin[2,1],u1,2),u2,0);

> quad[2,2]:=coeff(coeff(ODE_nonlin[2,1],u1,1),u2,1);

> quad[2,3]:=coeff(coeff(ODE_nonlin[2,1],u1,0),u2,2);

quad1,1 := 0

quad1,2 := Ψ01,2 ω β µ p1(cos(ωτ)− 1)

quad1,3 := Ψ01,2 β ω (ω p2− µ p1 sin(ωτ))

quad2,1 := 0

quad1,2 := Ψ02,2 ω β µ p1(cos(ωτ)− 1)

quad1,3 := Ψ02,2 β ω (ω p2− µ p1 sin(ωτ))

The necessary cubic coefficients are found in a similar way.

>cub:=array(1..2,1..4):

>cub[1,1]:=coeff(coeff(ODE_nonlin[1,1],u1,3),u2,0);

>cub[1,3]:=coeff(coeff(ODE_nonlin[1,1],u1,1),u2,2):

>cub[2,2]:=coeff(coeff(ODE_nonlin[2,1],u1,2),u2,1):

>cub[2,4]:=coeff(coeff(ODE_nonlin[2,1],u1,0),u2,3);

cub1,1 := Ψ01,2 β µ p1h2 11(0)(cos(ωτ)− 1)

cub1,3 := Ψ01,2 β
(
µ p2 cos(ω τ)ω2 − µ p2ω2 + p1µh1 12(−τ)ω − p1µh2 22(0)

+2 p2h2 12(0)ω − p1µh1 12(0)ω − p1µ sin(ω τ)h2 12(0)

+p1µ cos(ω τ)h2 22(0))

cub2,2 := −Ψ02,2 β (p1µh2 12(0)− p1µh1 11(−τ)ω + p1µ sin(ω τ)h2 11(0)

−p1µ cos(ω τ)h2 12(0) + p1µh1 11(0)ω − 2 p2h2 11(0)ω)

cub2,4 := −Ψ02,2 β
(
µ p2 sin(ω τ)ω2 + p1µ sin(ω τ)h2 22(0) + p1µh1 22(0)ω

−p1µh1 22(−τ)ω − 2 p2h2 22(0)ω)

Note that only the cubic terms depend on the hijk, as expected. The quantity
a is evaluated using the formula of eq. (42)



20 Sue Ann Campbell

a:=collect(simplify(1/8*(3*cub[1,1]+cub[1,3]+cub[2,2]

+3*cub[2,4])-1/(8*omega)*(quad[1,2]*(quad[1,1]

+quad[1,3])-quad[2,2]*(quad[2,1]+quad[2,3])

-2*quad[1,1]*quad[2,1]+2*quad[1,3]*quad[2,3]))),

[Psi0[1,2],Psi0[2,2]],distributed,factor);

a :=
1

64
β2ω p1 µ (cos (ω τ)− 1) (−ω p2 + µ p1 sin (ω τ))Ψ021,2 +

1

32
β2ω(

−ω2p2 2 − µ2p1 2 + 2ω p2 µ p1 sin (ω τ) + µ2p1 2 (cos (ω τ))
2
)
Ψ01,2Ψ02,2

+
1

8
β
(

p1 µ cos (ω τ) h2 22 (0)− 3µ p1 h2 11 (0)

+3µ p1 h2 11 (0) cos (ω τ) + 2 p2 h2 12 (0)ω − p1 µ h2 22 (0)

+p1 µ h1 12 (−τ)ω − p1 µ sin (ω τ) h2 12 (0)− µ p2 ω2 − p1 µ h1 12 (0)ω

+µ p2 cos (ω τ)ω2
)
Ψ01,2 −

1

64
β2ω p1 µ (cos (ω τ)− 1)

(
− ω p2

+µ p1 sin (ω τ)
)
Ψ022,2 −

1

8
β
(
− 2 p2 h2 11 (0)ω + p1 µ h2 12 (0)

−6 p2 h2 22 (0)ω + p1 µ h1 11 (0)ω + p1 µ sin (ω τ) h2 11 (0)

−p1 µ cos (ω τ) h2 12 (0)− p1 µ h1 11 (−τ)ω − 3 p1 µ h1 22 (−τ)ω

+3µ p2 sin (ω τ)ω2 + 3 p1 µ h1 22 (0)ω + 3 p1 µ sin (ω τ) h2 22 (0)
)
Ψ02,2

To get the final expression for a we need to substitute in the actual values for
Ψ0, hijk(0) and hijk(τ). The expression is very large, so we do not print it out.

afinal:=subs(Psi0_vals,simplify(subs(Csoln,

simplify(subs(hsoln0,hsolnt,a))))):

4 Discussion

In this chapter, we have shown how the symbolic algebra package MapleTMcan
be used to calculate the centre manifold for a delay differential equation at a
Hopf bifurcation. The commands involved are fairly simple, and thus it should
be fairly easily to adapt them to other computer algebra systems.

The emphasis of this chapter was on a system at a Hopf bifurcation. The
implementation for other bifurcations is similar, and just requires the modi-
fying the following parts:

1. The calculation of the basis functions for the centre eigenspace for the
original and transpose systems (Φ, Ψ).

2. The calculation of the quantity that determines the criticality. This de-
pends on the normal form for the bifurcation involved.

Also, as discussed in section 2, for some systems and some bifurcations, it
may not be necessary to compute the nonlinear terms of the centre manifold.

This chapter has focused on systems at a bifurcation. This means that our
predictions of the stability of the bifurcating limit cycle will only be valid in
some neighbourhood of the bifurcation point. To get predictions which valid
in a larger region, one can use the approach of parameter dependent centre
manifolds. For a general outline of this approach and some specific examples
for ordinary differential equations see (Guckenheimer and Holmes, 1983, p.
134) or (Wiggins, 1990, p. 198). For the application of the approach to DDEs
with a Hopf bifurcation see Faria and Magalhães (1995b) or Qesmi et al.
(2006b). For the application of this approach to DDEs with a Bogdanov-
Takens singularity see Faria and Magalhães (1995a) or Qesmi et al. (2007).
For the application of this approach to delay differential equations with a
Fold-Hopf singularity see Qesmi et al. (2006a). Note that the papers of Qesmi
et al. have some discussion of the implementation of their algorithms in Maple.

Finally, we would note that there are other approaches for studying the
dynamics of a delay differential equation near a nonhyperbolic equilibrium
point. Perturbation techniques (multiple scales, Poincaré-Lindstedt, averag-
ing) have been used to study Hopf bifurcation by Campbell et al. (2006); Chow
and Mallet-Paret (1977); Das and Chatterjee (2002); Gopalsamy and Leung
(1996); Nayfeh (2008); Rand and Verdugo (2007); Stech (1979) and Wirkus
and Rand (2004). Often the computations for such methods are as extensive
than for the centre manifold, however the mathematical theory is more ap-
proachable. The Liapunov-Schmidt reduction has also been implemented for
delay differential equations, both numerically by Stech (1985a,b,c); Aboud
et al. (1988) and using symbolic algebra by Franke and Stech (1991). How-
ever, this method only determines existence of the bifurcating solutions. Some
other method must be used to determined stability. Stech (1985b,c) discuss
such a method.

